
Misura remota di temperatura e umidità relativa 
mediante Arduino e la piattaforma IoT ThingSpeak (a cura di F.Parisi)



ThingSpeak: introduzione

ThingSpeak è una piattaforma open source che semplifica notevolmente l’implementazione di dispositivi 
IoT (Internet of Thing).

ThingSpeak consente di salvare nel cloud misurazioni inviate da uno o più sensori remoti, gestiti da client 
hardware (es. Arduino, ESP8266, Raspberry PI ecc.) connessi a una rete locale oppure a Internet, e di 
visualizzarli in grafici temporali.

Le caratteristiche principali di ThingSpeak sono:

- invio dei dati dei sensori alla piattaforma, in tempo reale, mediante i protocolli HTTP e MQTT
- visualizzazione dei dati mediante una pagina web, ospitata sulla piattaforma stessa o 
  in un sito web personalizzato, mediante widget e grafici temporali

In aggiunta è possibile:

- analizzare i dati mediante il linguaggio Matlab
- creare dei compiti in base ai dati: su evento («react») o periodici («time control»)
- esportazione dei dati in un foglio di calcolo

Le entità di ThingSpeak: il Channel e il Field

• Un channel identifica univocamente una o più variabili, associate a sensori/attuatori,
relative a uno stesso punto di misurazione  (es. “Casa Montagna”, “Stazione rilevamento 1”, “Ufficio”, ecc)

• Un field è una delle variabili del channel. Esempio il channel «Stazione  rilevamento 1» 
potrà avere diversi field: temperatura, umidità, posizione servomotore ecc.

• I field hanno i seguenti nomi: Campo1, Campo2 ecc

• Ciascun channel può avere fino a 8 fields, che possono memorizzare qualsiasi tipo di variabile (numerico o 
stringa), più tre per la localizzazione del punto di misurazione (latitudine, longitudine, altitudine) e uno per lo stato.

• Un channel è identificato da un numero univoco (es. 72696, 72654 ecc)

• Un channel è associato a un account, sul quale è possibile creare fino a 250 canali (nella versione standard) 

• Al channel si può accedere in lettura e in scrittura:
- es. scrittura: uno o più sensori che inviano periodicamente le misurazioni al server ThingSpeak
- es. lettura: una visualizzazione temporale in una pagina web
- es. lettura: un client che legge i field di un altro channel

• L’accesso in scrittura è protetto da una password («WRITE API KEY»)

• L’accesso in lettura è generalmente pubblico, ma può comunque essere privato, proteggendolo mediante una 
password di lettura («READ API KEY») , che ovviamente è sempre diversa da quella di scrittura

• Il server ThingSpeak supporta anche TLS/SSL (che deve essere ovviamente supportato anche dal client)

Versione gratuita di ThingSpeak

E’ disponibile una versione free ,ma con alcune limitazioni:
- massimo 3mln di scritture all’anno (circa 8200 scritture/giorno)
- massimo numero di canali: 4
- intervallo di invio: non inferiore a 15s (per ciascun canale)
- numero di subscribe simultanei (MQTT): 3
- nessun supporto tecnico 

Per le altre versioni (a pagamento) consultare l’indirizzo:

https://thingspeak.com/prices

Esempio di visualizzazione di un channel pubblico, mediante un web browser

https://thingspeak.com/channels/357142/

Come inviare i dati al server ThingSpeak mediante HTTP GET

E’ possibile aggiornare i field del canale, mediante una richiesta HTTP GET del client (in questo caso rappresentato da 
Arduino), al server HTTP ThingSpeak, secondo il seguente formato:

http://api.thingspeak.com/update.json?api_key=  API_KEY_SCRITTURA  &field1=VAL1&field2=VAL2 …

a questi parametri possono essere aggiunti altri opzionali, fra i quali: latitudine, longitudine e altitudine («LATITUDE», 
«LONGITUDE», «ELEVATION».) 

Esempio di richiesta HTTP GET:

http://api.thingspeak.com/  update.json  ?api_key  =  GK1ZMZ25YIYLMBGX  &field1=22&field2=65&latitude=40.8317&longitude=14.3385

 

I parametri «field», «latitude», «longitude», «altitude» se non specificati verranno impostati dal server a un valore nullo. 
Altri, invece, se non specificati verranno impostati dal server.

Ovviamente la richiesta andrà temporizzata (nel software del client) in modo da inviare periodicamente i dati aggiornati 
ogni Δt secondi.

N.B. a tal proposito, si ricorda che la versione free di ThingSpeak consente aggiornamenti solo ad intervalli di tempo 
superiori ai 15s (impostando tempi inferiori, il server ritorna un errore)

Come i dati inviati vengono salvati sulla piattaforma ThingSpeak

La richiesta HTTP inviata dal client verrà interpretata dal server e inserita in un record di un database. Ogni record è 
formato dai campi descritti nella successiva tabella. I parametri non esplicitati nella richiesta verranno automaticamente 
impostati dal server (es. il Channel ID e il TIMESTAMP) oppure a un valore nullo se non esplicitati diversamente. 
Esempio del record relativo alla richiesta vista in precedenza, con alcuni possibili valori dei campi:

channel_id 769027 ID del canale (determinato dalla API KEY)

created_at "2019-04-12T06:41:44Z" TIMESTAMP :data,ora,min,sec (in formato UTC ) dell’inserimento del dato

entry_id 186 Numero progressivo dell’inserimento

field1 "22" Valore passato dalla richiesta (tipo: intero, float o stringa max 255 caratteri)

field2 "65" Valore passato dalla richiesta (tipo: intero, float o stringa max 255 caratteri

field3 null Valore non passato dalla richiesta

field4 null “            “          “         “         “

field5 null “            “          “         “         “

field6 null “            “          “         “         “

field7 null “            “          “         “         “

field8 null “            “          “         “         “

latitude 40.8317 Latitudine, passato dalla richiesta (tipo:float )

longitude 14.3385 Longitudine, passata dalla richiesta (tipo: float)

elevation null Altitudine; valore non passato  nella richiesta (tipo: float)

status null
Campo a disposizione per descrizione canale, stato del canale (online/offline), commenti 
ecc (tipo: stringa max 255 caratteri); valore non passato dalla richiesta

Lab 1) uso della piattaforma ThingSpeak: creare un account gratuito 

• Aprire l’URL https://thingspeak.com e fare click su Registrati

• Compilare il form:
- indirizzo di email valido e attivo
- nome e cognome

• Spuntare la casella Use this email for my MathWorks Account

Premere Continue e attendere l’arrivo dell’email di conferma:

• Click sul link email (conferma creazione account MathWorks)

• Completare la registrazione con i dati richiesti

• Fare il Sign in sulla pagina https://thingspeak.com 

Lab 1) uso della piattaforma ThingSpeak: creare e configurare un channel (1)

Una volta entrati nel proprio account (“SIGN-IN”) 

Per creare un nuovo canale premere sul pulsante «NEW CHANNEL»

Lab 1) uso della piattaforma ThingSpeak: creare e configurare un channel (2)

Configurare quindi il canale, inserendo i campi necessari. Nel nostro caso  ci sono solo due 2 campi: 
TEMPERATURA è UMIDITA’ RELATIVA. 

Lab 1) uso della piattaforma ThingSpeak: creare e configurare un channel (3)

Ricordarsi di salvare il canale creato, premendo il pulsante SAVE (in fondo alla pagina)

Lab 1) uso della piattaforma ThingSpeak: visualizzare la lista dei channel creati 

Dal menù Canali  è possibile visualizzare l’elenco dei channel creati. Esempio:

La presenza del lucchetto chiuso indica che il canale è visualizzabile con accesso protetto (privato); 
viceversa il lucchetto aperto, indica che il canale è visualizzabile da tutti (accesso pubblico)

Lab 1) uso della piattaforma ThingSpeak: rendere pubblica la visualizzazione di un channel

Il menù “Sharing” consente di impostare la condivisione della pubblicazione: privata, pubblica (chiunque) 
o con altri utenti ThingSpeak (max 3 utenti). La visualizzazione consigliata maggiormente usata è quella  
pubblica:

Lab 1) uso della piattaforma ThingSpeak: configurare  la visualizzazione del channel (1)

Dal menù “I miei canali” selezionare il canale creato e aggiungere le visualizzazioni, nel nostro caso due 
diagrammi temporali, uno per la temperatura e un altro per l’umidità relativa.

13



Lab 1) uso della piattaforma ThingSpeak: configurare  la visualizzazione del channel (2)

Lab 1) uso della piattaforma ThingSpeak: visualizzare la chiave API di scrittura

La chiave API di scrittura serve come autenticazione per la scrittura dei dati sulla piattaform. Quella di 
lettura, invece, serve solo se optiamo per una visualizzazione privata. Sono entrambe visualizzabili nel 
menù “CHIAVI API”.

Lab 2) inviare i dati a ThingSpeak con Arduino ed Ethernet Shield: cosa serve

Materiale occorrente, per una stazione di rilevamento:

1 x Arduino Uno 1 x Ethernet Shield W5100  (*) 1 x Sensore temperatura/umidità 
DHT11

Librerie Arduino da installare, tramite «Gestione librerie» oppure manualmente:

• DHT
http://downloads.arduino.cc/libraries/github.com/adafruit/DHT_sensor_library-1.3.4.zip

• ThingSpeak
http://downloads.arduino.cc/libraries/github.com/mathworks/ThingSpeak-1.5.0.zip

* collegato a un router internet

Lab 2) inviare i dati a ThingSpeak con Arduino ed Ethernet Shield: schema a blocchi

E’ possibile ipotizzare un insieme di stazioni/punti di rilevamento: ogni stazione è composta da Arduino+Shield Eth.
+DHT11 ed è associata a un channel. Il numero max di channel è 4 nella versione free, ma può aumentare fino a 250 
(licenza standard).

Lab 2) inviare i dati a ThingSpeak con Arduino ed Ethernet Shield: lo schema di montaggio

Il pin DATA del sensore DHT11 verrà collegato al pin D3. Non è visibile nello schema la shield ethernet.

Lab 2) inviare i dati a ThingSpeak con Arduino ed Ethernet Shield: sviluppo del software (1)

Prima di scrivere lo sketch, occorre avere sotto mano le seguenti informazioni:

- Il channel ID , che si intende usare (per ciascun punto di rilevamento)
   e la relativa API KEY di scrittura (Write API Key)

- l’esatta corrispondenza fra i field e le variabili associate ai sensori/attuatori, 
  così come è stata definita nella configurazione del canale

  (es. Field1: Temperatura, Field2: Umidità relativa, ecc)

Si userà la libreria “ThingSpeak”, che semplifica notevolmente l’invio dei dati HTTP/GET al server 
ThingSpeak mediante Arduino.

Le funzioni, che consentono di impostare i campi di un canale, sono riportate si seguito

Lab 2) inviare i dati a ThingSpeak con Arduino ed Ethernet Shield: sviluppo del software (2)

setField( n, val )

Imposta il campo numero n (es.1, 2,3,…,8) al valore val, che può essere un numero (tipo intero 16-bit/32-bit o float 16-
bit) oppure una stringa (max 255 caratteri). Esempi:

ThingSpeak(1, 22.5);
ThingSpeak(1, 22);
ThingSpeak(2, “Hello World”)

setLatitude (val)

setLongitude( val)

setElevation (val)

Impostano l’altitudine, longitudine, altitudine al valore val (tipo float 16-bit) –  Vanno invocate (opzionalmente) dopo la 
impostazione dei campi. Esempi:

ThingSpeak.setLatitude(40.8317);
ThingSpeak.setLongitude(14.3385)

setStatus (val)

Imposta lo stato al valore val (stringa max 255 char). Esempio:

ThingSpeak.setStatus(“OFF”)

Lab 2) inviare i dati a ThingSpeak con Arduino ed Ethernet Shield: sviluppo del software (3)

Una volta definiti i campi e altri parametri opzionali, si inviano al server ThingSpeak, con la funzione writeFields() 

writeFields(channel_id, write_apikey);

Vengono “scritti sul canale”  anche i campi di di altitudine, latitudine, longitudine, status, eventualmente definiti in 
precedenza, sul canale channel_id,  con la chiave write_apikey.  Esempio d’uso:

int x = ThingSpeak.writeFields(769027, "GK1CMZ25YIYLMBGA");

Il valore x , ritornato dalla funzione di scrittura è un codice di stato, che descrive se la scrittura è andata a buon fine o ci 
sono stati dei problemi.

Alcuni codici di stato sono riportati di seguito:

CODICE Descrizione

200 OK /Success

400, 404 Chiave API scrittura non corretta o indirizzo server ThingSpeak non valido

-301 Errore connessione a ThingSpeak (es.instradamento di rete fallito)

-201 E’ stato specificato un numero di field non valido

-304 Timeout nella risposta dal server

-101 Stringa campo troppo lunga (maggiore di 255 caratteri)

Per l’elenco completo dei codici di stato: https://it.mathworks.com/help/thingspeak/error-codes.html

Lab 2) inviare i dati a ThingSpeak con Arduino ed Ethernet Shield: sviluppo del software (4)

E’ possibile, facoltativamente, rendere sicuro l’invio di dati, usando così la connessione al server HTTPS invece di HTTP 
(il client deve anche esso supportare TLS/SSL, come, per esempio: MKR1000, WiFI101 Shield , ESP8266, 
MKRGSM1400)

Aprire con un editor testo il file ThingSpeak.h   , posto nella directory:

C:\Users\nomeutente\Documents\Arduino\libraries\ThingSpeak\src

e modificare la seguente riga:

#define THINGSPEAK_PORT_NUMBER 80

in

#define THINGSPEAK_PORT_NUMBER 443

Dopo questa modifica, fare un nuovo upload dello sketch

Lab 2) inviare i dati a ThingSpeak con Arduino ed Ethernet Shield: sviluppo del software (5)

Lo sketch, relativo alla stazione 1 dell’esempio, sarà il seguente:

#include "ThingSpeak.h"
#include <SPI.h>
#include <Ethernet.h>
#include "DHT.h"

DHT dht (3, DHT11);   // pin D3, sensore DHT11
int rh;               // umidità relativa in percentuale
int temp;             // temperatura in gradi Celsius

// Dati accesso ThingSpeak:
const unsigned long channel_id = 769027;        // sostituire con l’id. del canale
const char* write_apikey = "GK1CMZ25YIYLMBGA";  // e con la Write API key 

// Definisce indirizzo MAC e indirizzo IP host, gateway, DNS, subnet mask:
byte mac[] = {0xDE, 0xAD, 0xBE, 0xEF, 0xFE, 0xED};
IPAddress ip  (192, 168, 8, 171);  // DNS/GW: 192.168.8.1 – SM:255.255.255.0
EthernetClient client;  // Arduino funziona da client Ethernet

void setup() {  
  Serial.begin(9600);       // iniz.funz. porta seriale (per i msg di debug)
  dht.begin();              // inizializza funzionamento sensore
  Ethernet.begin(mac, ip);  // inizializza  funzionamento interfaccia di rete
  Serial.print("IP del client: ");  //stampa debug su monitor seriale
  Serial.println(Ethernet.localIP());
  ThingSpeak.begin(client);         // inizializza client ThingSpeak
}

void loop() {

  // legge umidità e temperatura dal DHT
  rh = dht.readHumidity();
  temp = dht.readTemperature(); 

  // imposta i campi con i valori letti, secondo l’ordine del canale
  ThingSpeak.setField(1, temp);
  ThingSpeak.setField(2, rh);

  // scrive i campi sul channel e ritorna codice stato
  int x = ThingSpeak.writeFields(channel_id, write_apikey);

  // 200 codice OK
  if (x==200){
    Serial.println(F("Canale aggiornato con successo"));
  }
  else {
    Serial.print(F("Problemi nell'aggiornamento del canale. Codice errore: "));
    Serial.println(x);
  }
  delay(30000); // aspetta 30s
}

